NAG Library Routine Document

F04AGF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

F04AGF calculates the approximate solution of a set of real symmetric positive-definite linear equations with multiple right-hand sides, $AX = B$, where A has been factorized by F03AEF.

2 Specification

```
SUBROUTINE F04AGF(N, IR, A, LDA, P, B, LDB, X, LDX)
INTEGER           N, IR, LDA, LDB, LDX
double precision  A(LDA,N), P(N), B(LDB,IR), X(LDX,IR)
```

3 Description

To solve a set of real linear equations $AX = B$ where A is symmetric positive-definite, F04AGF must be preceded by a call to F03AEF which computes a Cholesky factorization of A as $A = LL^{\mathrm{T}}$, where L is lower triangular. The columns x of the solution X are found by forward and backward substitution in $Ly = b$ and $L^{\mathrm{T}}x = y$, where b is a column of the right-hand sides.

4 References

Wilkinson J H and Reinsch C (1971) *Handbook for Automatic Computation II, Linear Algebra* Springer–Verlag

5 Parameters

1: N – INTEGER *Input*

On entry: n, the order of the matrix A.

Constraint: N $\geq$ 1.

2: IR – INTEGER *Input*

On entry: r, the number of right-hand sides.

3: A(LDA,N) – ***double precision*** array *Input*

On entry: the upper triangle of the n by n positive-definite symmetric matrix A, and the subdiagonal elements of its Cholesky factor L, as returned by F03AEF.

On exit: is used as internal workspace, but is restored on exit.

4: LDA – INTEGER *Input*

On entry: the first dimension of the array A as declared in the (sub)program from which F04AGF is called.

Constraint: LDA $\geq$ N.

5: P(N) – ***double precision*** array *Input*

On entry: the reciprocals of the diagonal elements of L, as returned by F03AEF.

On exit: is used as internal workspace, but is restored on exit.

6: B(LDB,IR) – ***double precision*** array *Input*

On entry: the n by r right-hand side matrix B. See also Section 8.

7: LDB – INTEGER *Input*

On entry: the first dimension of the array B as declared in the (sub)program from which F04AGF is called.

Constraint: LDB $\geq$ N.

8: X(LDX,IR) – ***double precision*** array *Output*

On exit: the n by r solution matrix X. See also Section 8.

9: LDX – INTEGER *Input*

On entry: the first dimension of the array X as declared in the (sub)program from which F04AGF is called.

Constraint: LDX $\geq$ N.

6 Error Indicators and Warnings

None.

7 Accuracy

The accuracy of the computed solutions depends on the conditioning of the original matrix. For a detailed error analysis see page 39 of Wilkinson and Reinsch (1971).

8 Further Comments

The time taken is approximately proportional to n^2r.

9 Example

This example solves the set of linear equations $AX = B$ where

$$A = \begin{pmatrix} 5 & 7 & 6 & 5 \\ 7 & 10 & 8 & 7 \\ 6 & 8 & 10 & 9 \\ 5 & 7 & 9 & 10 \end{pmatrix} \quad \text{and} \quad B = \begin{pmatrix} 23 \\ 32 \\ 33 \\ 31 \end{pmatrix}.$$

9.1 Program Text

```
*     F04AGF Example Program Text
*     Mark 14 Revised. NAG Copyright 1989.
*     .. Parameters ..
      INTEGER          NMAX, IR, LDA, LDB, LDX
      PARAMETER        (NMAX=8,IR=1,LDA=NMAX,LDB=NMAX,LDX=NMAX)
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
*     .. Local Scalars ..
      DOUBLE PRECISION D1
      INTEGER          I, ID, IFAIL, J, N
*     .. Local Arrays ..
      DOUBLE PRECISION A(LDA,NMAX), B(LDB,IR), P(NMAX), X(LDX,IR)
*     .. External Subroutines ..
      EXTERNAL         F03AEF, F04AGF
*     .. Executable Statements ..
      WRITE (NOUT,*) 'F04AGF Example Program Results'
```

```
*     Skip heading in data file
      READ (NIN,*)
      READ (NIN,*) N
      WRITE (NOUT,*)
      IF (N.GT.0 .AND. N.LE.NMAX) THEN
         READ (NIN,*) ((A(I,J),J=1,N),I=1,N), ((B(I,J),J=1,IR),I=1,N)
         IFAIL = 1
*
*        Cholesky decomposition
         CALL F03AEF(N,A,LDA,P,D1,ID,IFAIL)
*
         IF (IFAIL.EQ.0) THEN
*
*           Approximate solution of linear equations
            IFAIL = 1
            CALL F04AGF(N,IR,A,LDA,P,B,LDB,X,LDX)
*
            WRITE (NOUT,*) ' Solution'
            DO 20 I = 1, N
               WRITE (NOUT,99998) (X(I,J),J=1,IR)
   20       CONTINUE
         ELSE
            WRITE (NOUT,99997) IFAIL
         END IF
      ELSE
         WRITE (NOUT,99999) 'N is out of range: N = ', N
      END IF
*
99999 FORMAT (1X,A,I5)
99998 FORMAT (1X,8F9.4)
99997 FORMAT (1X,' ** F03AEF returned with IFAIL = ',I5)
      END
```

9.2 Program Data

```
F04AGF Example Program Data
  4
    5    7    6    5
    7   10    8    7
    6    8   10    9
    5    7    9   10
   23   32   33   31
```

9.3 Program Results

```
 F04AGF Example Program Results

  Solution
    1.0000
    1.0000
    1.0000
    1.0000
```